In questo documento presentiamo i comandi nel linguaggio SINTASSI relativi alle procedure descritte nel capitolo 2. Si tratta di opzioni che consentono di completare lo spettro delle possibilità di analisi offerte da SPSS. 

Importare un file dati in formato testo con i comandi SINTASSI

E’ possibile importare un file testo (in formato variabile o fisso) in SPSS utilizzando i comandi scritti nel linguaggio di programmazione di SPSS. La procedura utilizzata è quella definita dal comando DATA LIST: descriviamola brevemente considerando l’importazione dei due file di esempio che possono essere scaricati dal sito (www.lededizioni.it).

Importazione di un file in formato fisso

Il file “esempio.dat” è costituito dalle seguenti 3 variabili: Numero d’ordine del soggetto (colonne 1-3), gruppo (colonna 4), scala (colonne 5-6). Di seguito presentiamo una breve sezione del file.

	  1142

  3169

........

156342

164342




Le linee di sintassi per leggere il file sono le seguenti: 

	DATA LIST FILE ‘C:\ESEMPIO.DAT’ RECORDS=1

/1 NORD 1-3 GRUPPO 4 SCALA 5-6.




Mandando in esecuzione queste linee di programma (nel modo che abbiamo specificato nel primo capitolo) si otterrà il seguente output:

	Data List will read 1 records from C:\ESEMPIO.DAT

Variable   Rec   Start     End         Format

NORD         1       1       3         F3.0

GRUPPO       1       4       4         F1.0

SCALA        1       5       6         F2.0




Il file è stato correttamente importato; le etichette delle colonne nell’output stanno ad indicare: il nome della variabile (Variable), il record ovvero la riga nel file originario in cui è contenuta la variabile (Rec), la colonna di inizio della variabile (Start), la colonna finale della variabile (End) e il formato Fortran della variabile (Format). Il formato della prima variabile va letto nel modo seguente: si tratta di una numero reale (F) che ha 3 cifre intere (3) e nessuna cifra decimale (0). Per una completa definizione delle opzioni relative al formato Fortran si rimanda ai manuali del programma. 

Consideriamo ora un altro esempio di importazione di dati in formato fisso, esaminando un file nel quale i dati di ogni soggetto sono riportati su più di una riga, e che prevede variabili con cifre decimali. Il file è “ESEMPIO2.DAT”, ed è costituito dalle seguenti variabili: 

Riga 1: Numero d’ordine del soggetto (colonne 1-4), classe (colonna 4)

Riga 2: Scala 1 (colonne 1-4), Scala 2 (colonne 5-8). 

Di seguito presentiamo una breve sezione del file.

	7014

2,002,00

 7024

2,201,73

........

11854

2,402,40

14234

2,401,67




Le variabili Scala 1 e Scala 2 prevedono 1 cifra intera, la virgola e 2 cifre decimali. 

Le linee di sintassi per leggere il file sono le seguenti: 

	DATA LIST FILE ‘C:\ESEMPIO2.DAT’ RECORDS=2

/1 NORD 1-4 CLASSE 4 

/2 SCALA_1 1-4(2) SCALA_2 5-8(2).




Ogni riga del file originale corrisponde ad un “record” nel comando DATA LIST, e questo viene indicato con il carattere “/” (abbiamo scritto anche il numero del record nella nostra sintassi, /1 e /2, ma questo può essere anche tralasciato). Le cifre decimali vengono segnalate aprendo una parentesi subito dopo la colonna finale della variabile, specificando il numero di cifre decimali e chiudendo la parentesi. La virgola non va segnalata, anche se la posizione della variabile nel file la considera. Infatti, per le due variabili con cifre decimali bisogna indicare che la loro lunghezza complessiva è pari a 4 colonne ovvero 4 caratteri (in queste 4 colonne sono comprese: la cifra intera, la virgola, le due cifre decimali).

Mandando in esecuzione queste linee di programma si otterrà il seguente output:

	Data List will read 2 records from C:\ESEMPIO2.DAT

Variable   Rec   Start     End         Format

NORD         1       1       4         F4.0

CLASSE       1       4       4         F1.0

SCALA_1      2       1       4         F4.2

SCALA_2      2       5       8         F4.2




L’interpretazione è la stessa del file precedente: le novità sono rappresentate dal fatto che ora le variabili sono contenute, nel file originale, in 2 record diversi (vedi colonna Rec), e che ci sono due variabili (SCALA_1 e SCALA_2) che hanno 2 cifre decimali (vedi colonna Format). 

Importazione di un file in formato variabile

Il file “ese_efa_1.dat” è costituito dalle seguenti variabili: 

- numero d’ordine del soggetto, 

- schieramento politico votato alle ultime elezioni, 

- 6 item relativi ad una scala di personalità utilizzata per rilevare la personalità del candidato premier. 

Di seguito presentiamo una breve sezione del file:

	101
1
7
7
7
5
6
6

102
2
7
7
4
2
4
2

103
1
7
5
6
5
7
6

..........................................

20806
3
7
6
6
6
4
4

99993
1
5
5
6
4
5
4




Diversamente dai file esaminati negli esempi precedenti, si tratta di un file in cui ogni coppia di variabili è separata da un carattere di tabulazione. Le linee di sintassi per leggere il file sono le seguenti: 

	DATA LIST FILE 'C:\ESE_EFA_1.DAT' free

/nord schier it1 it2 it3 it4 it5 it6.




Essendo il file in formato libero non viene riportata nessuna informazione in merito alla posizione delle variabili nel data file originale. Se invece di utilizzare il carattere di tabulazione si utilizza un altro carattere per separare le variabili, il carattere va segnalato nel DATA LIST nel modo seguente: 

	DATA LIST FILE 'C:\ESE_EFA_2.DAT' free (",")

/nord schier it1 it2 it3 it4 it5 it6.




In questo esempio abbiamo considerato la virgola come separatore (se si utilizza un altro carattere, questo va sostituito alla virgola). Il comando precedente consente di leggere un file come quello successivo:

	101,1,7,7,7,5,6,6

102,2,7,7,7,5,6,6

.................




E’ possibile importare file testo in formato variabile che contengono numeri decimali. Consideriamo il seguente file (ESEMPIO4.DAT):

	 701    4    2    2

 702    4    2,20    1,73

 703    4    2,20    2,20

 704    4    2    2,07

 705    4    2    2

...........................

1185    4    2,40    2,40

1423    4    2,40    1,67




E’ facile verificare che le variabili non sono allineate nelle colonne in corrispondenza di una stessa posizione per ogni soggetto (il file è in formato variabile/free). Inoltre le ultime due variabili presentano una virgola e due cifre decimali. In realtà il DATA LIST per questo file è molto semplice essendo il seguente:

	DATA LIST FILE 'C:\ESEMPIO4.DAT' free

/NORD CLASSE SCALA_1 SCALA_2.




La sola indicazione “free” basta per segnalare la presenza di un file in formato variabile, mentre non è necessario segnalare le variabili che presentano cifre decimali perché il programma provvederà automaticamente a riconoscerle. Il risultato di questa importazione è presentato nella figura 1. Ovviamente, se nel file dati sono presenti valori decimali non è possibile utilizzare la virgola come carattere separatore dei dati. 

Figura 1. File in formato free importato come data file SPSS

________________________________________________________________________________

________________________________________________________________________________

Importazione di file testo in formato libero

Riportiamo di seguito i comandi SINTASSI relativi all’importazione dei dati in formato libero.  

	*Sintassi relativa alla procedura di importazione di dati testo in formato libero*.

GET DATA  /TYPE = TXT

 /FILE = 'C:\libro.spss\libro spss\ese_efa_1.dat'

 /DELCASE = LINE

 /DELIMITERS = "\t"

 /ARRANGEMENT = DELIMITED

 /FIRSTCASE = 1

 /IMPORTCASE = ALL

 /VARIABLES =

 V1 F3.2

 V2 F1.0

 V3 F1.0

 V4 F1.0

 V5 F1.0

 V6 F1.0

 V7 F1.0

 V8 F1.0

 .

CACHE.

EXECUTE.




Importazione di file testo in formato fisso

La sintassi per questa procedura, ottenuta specificando “SI” nella finestra di dialogo in figura 16 nella sezione “Incollare la sintassi ?”, è fornita di seguito. 

	* Sintassi per importare un file in formato fisso*.

GET DATA  /TYPE = TXT

 /FILE = 'C:\libro.spss\libro spss\esempio.dat'

 /FIXCASE = 1

 /ARRANGEMENT = FIXED

 /FIRSTCASE = 1

 /IMPORTCASE = ALL

 /VARIABLES =

 /1  id 0-2 F3.0

 gruppo 3-3 F1.0

 scala 4-5 F2.0

 .

CACHE.

EXECUTE.




Importazione di file in formato Excel
Di seguito viene presentata la sintassi relativa al comando “Apri/Dati” necessaria per importare in SPSS un file in formato Excel. 

	* Sintassi per importare un file in formato Excel*.

GET DATA /TYPE=XLS

   /FILE='C:\MTMM_BFQC\correlazioni.xls'

   /SHEET=name 'Foglio1'

   /CELLRANGE=full

   /READNAMES=on .




Salvare ed esportare un file dati
Di seguito viene presentata la sintassi relativa al comando “Salva con nome” in cui sono state escluse 2 variabili dal file originale. 

	* Comandi per salvare un file in formato sav*.

SAVE OUTFILE='C:\prova.sav'

  /DROP=sex eta 

  /COMPRESSED.




Le linee di sintassi relative all’opzione “salva come ASCII fisso”  a questa operazione sono le seguenti: 

	WRITE OUTFILE='C:\libro.spss\libro spss\prova.dat'

  TABLE

  /nord sex eta sch agg2 agg8 agg9 agg11 agg20 agg21  .

EXECUTE.




Le linee di sintassi relative all’opzione “salva come Tab delimitato”  sono le seguenti: 

	SAVE TRANSLATE OUTFILE='C:\libro.spss\libro spss\prova.dat'

  /TYPE=TAB /MAP /REPLACE /FIELDNAMES.




Le linee di sintassi relative all’opzione “salva come Portabile SPSS”   sono le seguenti: 

	EXPORT OUTFILE='C:\libro.spss\libro spss\prova.por'.




Le linee di sintassi relative all’opzione “salva come Excel 2.1”   sono le seguenti: 

	SAVE TRANSLATE OUTFILE='C:\libro.spss\libro spss\prova.xls'

  /TYPE=XLS /VERSION=2 /MAP /REPLACE /FIELDNAMES .




Aggiungi casi/Aggiungi variabili
Le linee di sintassi relative all’operazione “Aggiungi casi” sono le seguenti: 

	ADD FILES /FILE=* 

/FILE='C:\eterotrait.bis.sav'.




Le linee di sintassi relative all’operazione “Aggiungi variabili” sono le seguenti: 

	MATCH FILES /FILE=*

 /FILE='C:\libro.spss\libro spss\SPSS FILES\ana_fatt.sav'

 /RENAME (eta sex = d0 d1)

 /BY nord

 /DROP= d0 d1.




Aggregazione
Le linee di sintassi relative all’operazione di aggregazione sono le seguenti: 

	AGGREGATE

  /OUTFILE='C:\aggr.sav'

  /BREAK=politico

  /agg_1_1 = MEAN(agg_1) 

  /agg_2_1 = MEAN(agg_2) 

  /agg_1_2 = SD(agg_1) 

  /agg_2_2 = SD(agg_2) 

  /agg_1_3 = MIN(agg_1) 

  /agg_2_3 = MIN(agg_2) 

  /agg_1_4 = MAX(agg_1) 

  /agg_2_4 = MAX(agg_2)

  /N_BREAK=N.




Selezione dei casi

Le linee di sintassi relative alle operazioni di selezione dei casi sono le seguenti: 

	*** casi filtrati ****.

COMPUTE filter_$=(nord > 50).

VARIABLE LABEL filter_$ 'nord > 50 (FILTER)'.

VALUE LABELS filter_$  0 'Non selezionato' 1 'Selezionato'.

FORMAT filter_$ (f1.0).

FILTER BY filter_$.

**** casi cancellati

SELECT IF(nord > 50).

**** casi filtrati casualmente .

COMPUTE filter_$=(uniform(1)<=.50).

VARIABLE LABEL filter_$ 'Approssimativamente 50 % di casi (CAMPIONE)'.

FORMAT filter_$ (f1.0).

FILTER BY filter_$.

EXECUTE .

**** casi selezionati casualmente e cancellati

SAMPLE .50.

**** filtro dei primi 500 casi

use  1 thru  500  .

*** selezione dei primi 500 casi e cancellazione dei restanti

use  1 thru  500 /permanent .




Calcolo di un variabile

Le linee di sintassi relative alle operazioni di calcolo non condizionato e condizionato di una variabile sono le seguenti: 

	*** calcolo non condizionato **.

COMPUTE apertura = agg_1 + agg_7 + agg_14 + agg_17 + agg_18 .

*** calcolo condizionato **.

IF (apertura >=  20) apert_f = 2 .

IF (apertura <   20) apert_f = 1 .



Ricodifica di una variabile 

Le linee di sintassi relative alle operazioni di ricodifica di una variabile sono le seguenti: 

	**** ricodifica nella stessa variabile .

RECODE   politico  (1=2)  (2=3)  .

**** ricodifica in una variabile differente.

RECODE   politico   (1=1)  (2=2)  INTO  polit_2 .

**** ricodifica in una variabile differente alfanumerica.

STRING sesso (a8).

RECODE   sex   (1='m')  (2='f')  INTO  sesso .

**** ricodifica in una stessa variabile alfanumerica.

RECODE

 sesso   ('m'='maschio')  ('f'='femmina').

**** ricodifica condizionale.

DO IF (sex  = 1) .

RECODE   politico  (1=2)  (2=3)  .

END IF .




Classifica variabili/ Classifica Casi
Le linee di sintassi relative all’operazione “Classifica variabili”  sono le seguenti: 

	RANK

  VARIABLES = NETA

  /NTILES(4)

  /PRINT = NO

  /TIES = MEAN .




Le linee di sintassi relative all’operazione “Classifica Casi” sono le seguenti: 

	**** ordinamento semplice.

RANK

  VARIABLES=agg_1  (A) /RANK /PRINT=YES

  /TIES=MEAN .

**** ordinamento separato rispetto a una variabile.

RANK

  VARIABLES=agg_1  (A) BY sex  /RANK /PRINT=YES

  /TIES=MEAN .




Ricodifica automatica
Le linee di sintassi relative all’operazione “Ricodifica automatica” sono le seguenti: 

	AUTORECODE

  VARIABLES=politico  /INTO polit_2

  /PRINT.



